ENGINEERING WITH EXCEL

A Practical Guide to Using Excel for Advanced Engineering Calculations

CHARLIE YOUNG, P.E.

Hi there!

Thanks so much for purchasing *Engineering with Excel*. I'm honored that you have chosen me to teach you how to use Excel for advanced engineering calculations.

Before you get started, I want to quickly explain what you can expect from this book.

As the subtitle says, this book is a "practical guide". That means it is intended to be used as a hands-on resource.

Most people (and I'm guessing that includes you) learn concepts best when they can apply it to a real-life situation.

That's why each section of every chapter in this book is a practical exercise for you to follow along with so that you can really learn the concepts and commit them to memory.

Along with your purchase of this book, you also received over 80 spreadsheets. These spreadsheets are used throughout the book as examples that you can follow.

You'll learn more by doing the exercises rather than just reading about them!

At the beginning of most sections is a reference to the spreadsheet used in that section. Open the file and follow along with the steps described in that section.

Later, if you need to come back and refresh your memory on a concept, you can always quickly brush up on a particular section.

Finally, the chapters in this book build on one another, so it's highly recommended that you work through the chapters in order.

That's all for now. Happy Excelling!



Charlie Young, P.E.

Contents

Chapter 1: Foundational Excel Functions for Engineering

Section 1: Using Functions

Functions are the backbone of performing engineering calculations in Excel. Functions are used to create formulas. There are 3 different ways that you can enter them.

As an example, let's say we have a number in our spreadsheet, and we need to find its square root. If you don't know the exact syntax, you can find it in the **Formula** tab. Since it's a math function, go to the **Math & Trig** functions menu:



Scroll down until you find the square root function (**SQRT**). When you click it, a dialog box will open. This box includes an explanation of the function and the arguments it takes:



Select the number that you want to use in your calculation and click OK.

Another way to search for functions is using the **Insert Function** button next to the formula bar (f_x on the right):

When you click the f_x button, you can enter a search term. After you find the correct function, click OK, and it will bring you to the same dialogue box as above. **Select** the cell that you want to use in the calculation and click OK.

These two methods of searching for formulas are helpful when you're starting out with Excel because they guide you through the creation of a formula by explaining what the function does and what arguments are necessary.

However, these methods involve lots of clicking and dialogue boxes, so they become a hindrance to productivity. Learning to type functions in will help increase your speed.

To use the square root function, click within a cell and type the **equals sign**, then start typing the name of the function. As soon as you start typing, Excel suggests a list of functions that start with that letter. As you type it narrows down the list. Simply typing **sq** will narrow it down to two choices:



A short description of the function appears as a tool tip. You can use the **Up** and **Down Arrow** keys to navigate within the list. Press the **Tab** key to choose a function, and it will fill in the rest of the function's name and the open parenthesis:

Below the cell, a tooltip box appears to lead you through the arguments that the function takes. This is helpful for complex functions that have multiple arguments. Of course, in the case of the square root function, it's pretty simple because there's only one argument (**number** in the tool tip). To add the argument, you can navigate to the number **4** using the arrow keys (**Up Arrow** in this case), or just click its cell. It's unnecessary to add the closing parenthesis. Just hit Enter.

Entering functions by typing is the fastest way to work with functions. This method will be used throughout the remainder of this book.

Section 2: Absolute and Relative Cell References

Worksheet: 01b Fan Airflow.xlsx

Inevitably, you will often reference other cells in your calculations. There are two different types of cell references in Excel: relative and absolute. This is an important concept to understand for entering formulas in Excel.

Relative cell references will change when a formula is copied to another cell. Absolute cell references will remain constant when a formula is copied to other cells.

As an example, see the simple table below. There are a series of values in column A. In cell B1, there's a formula with a **relative** reference to cell A1. If you **fill** that formula down (using the **fill handle** from Chapter 1, Section 7), the other cells in column B will refer to the cell directly to their left, **not** cell A1:



An **absolute** reference would maintain the reference to cell A1 in all of the cells:



For a table such as this, an absolute reference might not make sense, but there are many situations where it's necessary.

In Excel, absolute references are denoted using a **dollar sign** before the row and column designators. A quick way to turn a *relative* reference into one that is *absolute* is to use the **F4** key to toggle. The first time you type **F4**, the reference will become a fixed cell absolute reference, i.e. $A$1. This reference will always stay locked onto A1.

However, pressing **F4** a second time will allow you to fix the row reference while allowing the column to change (i.e. A$1). If you copy this reference, the column letter will be able to change (to A, B, C, etc.) like a relative reference, but it will always reference the first cell in the column.

Typing **F4** a third time will fix the column reference but allow the row to change (i.e. $A1).

The fourth time **F4** is pressed, it changes back to a relative reference.

Worksheet 01b contains an example that requires different types of references to complete the table. This is a fan airflow spreadsheet that uses fan affinity laws to calculate airflow as speed and fan diameter change based on a measured airflow with a known fan diameter and speed. The fan affinity laws state that the new airflow is proportional to the ratio of fan speeds (*n*) and the ratio of fan diameters (*d*) cubed:

$$q_2 = q_1\left(\frac{n_2}{n_1}\right)\left(\frac{d_2}{d_1}\right)^3$$

This formula can be used to calculate the airflow from a reference fan speed, fan diameter and airflow rate as the speed and fan diameter change. In Worksheet 01b, the reference fan speed, fan diameter and airflow are in a smaller table above.

Enter the formula into the first cell in the large table: **=C7*(B12/C5)*(C11/C6)^3** (see next page). Excel uses relative cell references by default, so use those for now.



Fan Airflow

Inputs		
Fan Speed	2000	RPM
Fan Diameter	20	in
Air Flow Rate	3000	CFM

Fan Affinity Law:

$q_2 = q1*(n_2 / n_1)(d_2 / d_1)^3$

Outputs							
Speed					Fan Diameter (in)		
(RPM)	15	16	17	18	19	20	21
1500	=C7*(B12/C5)*(C11/C6)^3						
1600							
1700							
1800							

This will calculate the expected airflow if a similar fan of smaller diameter was run at a lower speed. However, if you use the **fill handle** to fill this formula into the rest of the table, there will be errors in all the other cells:

Outputs							
Speed						Fan Diameter (in)	
(RPM)	15	16	17	18	19	20	21
1500	949	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!
1600	0	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!
1700	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!
1800	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!

This is because the references have moved. Absolute references are required here.

The three references to the original fan (in the small table) should stay fixed on a single cell. To do this, **select** the cell with the formula, **click** within the reference and type **F4** once. Do this for the references to cells C5, C6, and C7. This will add dollar signs before both the column letter and the row number, indicating that this reference will be constant.

The n_2 variable is the speed of the new fan. This should be allowed to vary, because we have a number of speeds in the column on the left side of the table. For this reference, it will be necessary to fix the *column*, but allow the *row* to vary. **Click** within the reference to cell B12 and type **F4** three times to obtain $B12.

The d_2 variable (C11) is the diameter of the new fan, located in the row along the top of the table. To lock onto this row and allow the column to vary, type **F4** twice to obtain C$11.

Fill this formula into the rest of the table by **dragging the fill handle** first to the right, then down.

To verify that the cell references are correct, **click** within any cell then within the formula bar. The reference cells will be highlighted with a colored box. Each cell should highlight the corresponding fan diameter at the top of the table and the corresponding speed on the left side.

Inputs		
Fan Speed	2000	RPM
Fan Diameter	20	in
Air Flow Rate	3000	CFM

Fan Affinity Law:

$$q_2 = q1*(n_2 / n_1)(d_2 / d_1)^3$$

Outputs							
Speed	Fan Diameter (in)						
(RPM)	15	16	17	18	19	20	21
1500	949	1152	1382	1640	1929	2250	2605
1600	1013	1229	1474	1750	2058	2400	2778
1700	1076	1306	1566	1859	G$11/$C$	2550	2952

Section 3: SUM, PRODUCT, QUOTIENT, and ABS

Worksheet: 01b Basic Math Functions

This section will cover a few basic mathematical functions: sum, product, quotient, and absolute value.

The **SUM** function will quickly add numbers together. We can manually add together the cells using the formula bar (below left), but the SUM function is faster (below right). Type **=SUM(** into the desired cell, **select** the cells to be added with the mouse and press **Enter**.



The **PRODUCT** function works in a similar fashion. Rather than manually multiplying each cell (below left), we can simply type **=PRODUCT(** and select the cells to be multiplied.



The **QUOTIENT** function is not exactly the same as dividing. When you manually enter in a formula to divide 5 by 2 using the forward slash **/**, Excel will return 2.5 by default. When you use the **QUOTIENT** function, Excel returns the integer 2, dropping the remainder. When using the **QUOTIENT** function, remember to separate the two input cells with a **comma** (**numerator,denominator**).



The **ABS** function returns the absolute value function of a given cell. If the input value is negative, it converts it to positive. If the input value is positive, it will be returned unchanged.



Section 4: Trigonometric Functions

Worksheet(s): 01d Trigonometric Functions

Excel contains a variety of trigonometric functions:

Basic functions	Hyperbolic functions	Inverse functions	Miscellaneous

SIN	SINH	ASIN	PI()
COS	COSH	ACOS	RADIANS
TAN	TANH	ATAN	DEGREES
SEC	SECH	ASINH	
CSC	CSCH	ACOSH	
COT	COTH	ATANH	

The inverse functions are those usually denoted with a superscript -1 in math (i.e. **ASIN** is the Excel function for $\sin^{-1}$). These will return an angle given a sine value (or cosine, tangent, etc.).

The "Miscellaneous" column contains functions that are useful in trigonometric calculations. **PI()** returns the value of π to 15 digits. **RADIANS** will convert a number from degrees into radians. **DEGREES** converts from radians into degrees.

It's important to note that Excel uses **radians** for all trigonometric functions. If you're working with angles in degrees, you'll have to convert into radians before inputting those values into one of these functions. You can use the **RADIANS** function directly, or you can use the formula:

$$\theta_{radians} = \left(\frac{\pi}{180}\right) * \theta_{degrees}$$

Let's say there's an angle in degrees in cell A1 of your worksheet. You can enter the above formula in the form: **=PI()/180*A1**. You can also insert that term directly into one of the trig functions. For example, to take the sine of cell A1, enter **=SIN(PI()/180*A1)** into the formula bar.

Worksheet 01d contains a spreadsheet to calculate the sine, cosine and tangent for 0 to 2π. Since the angles are already in radians, it's straightforward to do these calculations. In the first row, type **=SIN(** and select the cell containing the angle (**C5**). Repeat for the cosine **=COS(C5)** and tangent **=TAN(C5)**. To fill the rest of the table, **select** all three cells, **hover** over the **fill handle** and **double-click**.



The worksheet will automatically plot the data when you fill in the table. There are undefined values in the tangent at $\frac{\pi}{2}$ and $\frac{3\pi}{2}$ which appear as ######. You can delete these to clean up the plot.

To calculate the angle in degrees, enter the following into the first cell: **=DEGREES(C5)**. **Double-click** the **fill handle** to complete the table. As an exercise, let's use the angle in degrees as inputs for the sine, cosine and tangent columns. Delete the data in these three columns. In the sine column, enter in the formula: **=SIN(RADIANS(B5))**. The function to convert B5 to radians is nested within the SIN function. You could also use **=SIN(PI()/180*B5)** if you prefer to use the formula on the previous page.

Repeat for the cosine and tangent functions: **=COS(RADIANS(B5))** and **=TAN(RADIANS(B5))**, respectively. **Highlight** the three cells and **double-click** the **fill handle**. This gives us the same data as before, but eliminates the need for a separate radians column if your input data is in degrees.

Section 5: Inverse Trigonometric Functions

Worksheet: 01e Inverse Trig Functions

We'll use Worksheet 01e to practice using inverse trigonometric functions and conversions between radians and degrees. The functions ASIN, ACOS and ATAN can be used to find the inverse sine, cosine and tangent of the values in column B.

Inverse Trig Functions

	$\sin^{-1}$	$\cos^{-1}$	$\tan^{-1}$
-1.00	=ASIN(B5)		
-0.90			
-0.80			
-0.70			

Enter the formulas **=ASIN(B5)**, **=ACOS(B5)** and **=ATAN(B5)** in the columns for $\sin^{-1}$, $\cos^{-1}$ and $\tan^{-1}$, respectively. **Select** these three cells and **double-click** the **fill handle** to fill the columns. These three functions will return angles in radians:

Inverse Trig Functions

	$\sin^{-1}$	$\cos^{-1}$	$\tan^{-1}$
-1.00	-1.57	3.14	-0.79
-0.90	-1.12	2.69	-0.73
-0.80	-0.93	2.50	-0.67
-0.70	-0.78	2.35	-0.61

To display the angles in degrees, you can nest these functions inside of the **DEGREES** function. **Double-click** the first cell in the $\sin^{-1}$ column to edit it, and change the formula to **=DEGREES(ASIN(B5))**. Edit the $\cos^{-1}$ and $\tan^{-1}$ columns in the same way. **Select** the three edited cells and **double-click** the **fill handle**. Now the data will be in degrees:

Inverse Trig Functions

	$\sin^{-1}$	$\cos^{-1}$	$\tan^{-1}$
-1.00	-90.00	180.00	-45.00
-0.90	-64.16	154.16	-41.99
-0.80	-53.13	143.13	-38.66
-0.70	-44.43	134.43	-34.99

Section 6: LN, LOG, LOG10, EXP, SQRT, and FACT

Worksheets: 01f True Stress and Strain, 01g Sound Power Level, 01h Decaying Oscillation

Excel features many of the common mathematical functions that you'll encounter in engineering: logarithms, exponentials, square roots and factorials.

Mathematical Notation	Excel Notation
$\ln(x)$	LN(x)
$\log_n(x)$	LOG(x,n)
$\log_{10}(x)$	LOG10(x)
e^x	EXP(x)
$\sqrt{x}$	SQRT(x)
$n!$	FACT(n)

This section covers three example problems that use these functions. Worksheet 01f is set up to calculate true strain from engineering strain data collected on a tensile test machine. The equation for true strain is:

$$\epsilon_{tr} = \ln(1 + \epsilon_e)$$

where ϵ_e is the engineering strain found in column B. **Select** the first cell in the true strain column and enter the formula **=LN(1+B5)**. Press **Enter**, **select** that cell and **double-click** the **fill handle** to apply the formula to the rest of the column. This will allow you to see how the true stress and strain compare to the engineering stress and strain.



True Stress and Strain

Engineering Strain	Engineering Stress (MPa)
0.0000	0.0
0.0058	478.2
0.0080	498.8

True Strain	True Stress (Mpa)
=LN(1+B5)	0
	481
	503

Logarithms come up often when dealing with sound measurements. For instance, a logarithm is used to calculate the overall sound pressure from sound measurements at individual frequencies. In Worksheet 01g, the A-weighted sound pressure level can be calculated from measured sound levels using the equation:

$$L = 10 \log \sum 10^{L_i/10}$$

First, you'll have to calculate $10^{L_i/10}$ from the corrected sound value, L_i, in column E. Enter **=10^(E4/10)** in the first cell of column F. **Select** the first cell and double-click the **fill handle**. Next, we'll calculate L using the **SUM** function to get the sum of the values in column F. In the large cell next to "Sound Pressure," enter **=10*LOG(SUM(F4:F11))**. The **LOG** function assumes base ten unless you specify the base in a second argument, i.e. LOG(8,2) will find $\log_2 8$.

Corrected Value (dB)	10^Li/10
64.1	2570396
74.9	30902954
88.5	711213514
77.0	50582466
76.3	42266861
79.8	95060479
72.5	17782794
53.5	225944

=10*LOG(SUM(F4:F11))	dBA

Worksheet 01h contains an example using the exponential function. This function is used often for decaying oscillations. The ratio of successive oscillation amplitudes can be calculated if the damping ratio is known using the formula:

$$\frac{x_1}{x_2} = e^{\frac{2\pi\delta}{\sqrt{1-\delta^2)}}}$$

where x_1 is the first oscillation amplitude, x_2 is the second oscillation amplitude, and δ is the damping ratio. Worksheet 01h contains the first oscillation amplitude in cell C4 and the damping ratio in cell C5, so we can rearrange the equation to solve for the second oscillation amplitude. In the cell next to "Second Oscillation Amplitude," enter **=C4/EXP((2*PI()*C5)/SQRT(1-C5^2))**.

Result	
Second Oscillation Amplitude, x2 =	=C4/EXP((2*PI()*C5)/SQRT(1-C5^2))

Note that we've used three Excel functions here: **EXP** for the exponential, **PI()** for the value of π, and **SQRT** for the square root. Enclosing the exponent's numerator and denominator in parentheses helps ensure Excel reads our formula correctly – this is a good habit to adopt.

Section 7: Complex Number Functions

Worksheet: 01j Complex Numbers

If you do any engineering calculations that involve complex numbers, you'll probably appreciate the complex number functions that are built into Excel. Excel has many functions for working with complex numbers, but we'll only cover a few here.

Function	Description
COMPLEX	Creates a complex number
IMAGINARY	Extracts the imaginary coefficient from a complex number
IMREAL	Extracts the real coefficient from a complex number
IMDIV	Divides complex numbers
IMPRODUCT	Multiplies complex numbers
IMSUB	Subtracts complex numbers
IMSUM	Adds complex numbers

You can browse through the other functions for complex numbers by going to **Formulas** > **More Functions** > **Engineering**. Scroll down to find the functions that begin with "IM." Hover over each function to get a brief description. You can also see the Excel help for more information.

The **COMPLEX** function takes real and imaginary coefficients and creates a complex number, either in **x+yi** or **x+yj** form. The syntax is as follows:

> COMPLEX(real_num, i_num, [suffix])
> real_num: the real part of the complex number (x)
> i_num: the imaginary part of the complex number (y)
> suffix: (optional) suffix for the imaginary part – either "i" (default) or "j"

If you omit the suffix, Excel will use i by default. It's important to enclose "i" and "j" in double quotation marks in the suffix argument.

The **IMAGINARY** function can be used to find the imaginary part of an existing complex number – **y** in **x+yi** – and returns it to a cell. There's only one argument for this function

– the complex number. The **IMREAL** function will extract the real part of a complex number (**x**).

You can't directly add, subtract, multiply, or divide complex numbers in Excel using symbols (+, -, etc). To perform those operations with complex numbers, you'll need to use these special functions: **IMDIV**, **IMPRODUCT**, **IMSUB** and **IMSUM**.

A common example in engineering that uses complex numbers is an AC circuit. In Worksheet 01j, there's an example that calls for complex number arithmetic:



First, enter in the specified voltage (45+10j) as a complex number. The real part of the voltage is 45 – this will be the first argument. The imaginary part is 10, the second argument. We'll specify "j" in the third argument so as to not confuse "i" with current. Therefore, the entry for voltage will be **=COMPLEX(45,10,"j")**. Excel will display this as 45+10j. Do likewise for impedance: **=COMPLEX(3,4,"j")**.

We need to divide voltage by impedance to find the current, but it's not possible to simply divide the two complex numbers with the / operator. Use the **IMDIV** function. This function takes the numerator as its first argument, and the denominator as its second. Therefore, to find the current, enter **=IMDIV(C6,C7)**. Remember to separate the two arguments with a comma, not a slash.

This returns a result of 7-6j for the current.

Section 8: Rounding Functions

Worksheet: 01k Rounding

Excel has several functions for rounding numbers.

Function	Description
INT	Rounds down to the nearest integer
ROUND	Rounds the number to a specified number of digits
ROUNDUP	Rounds a number up, away from zero
ROUNDDOWN	Rounds a number down, toward zero
MROUND	Rounds to the nearest multiple specified

The **INT** function has only one argument, the number to be rounded. This function effectively rounds down to the nearest integer.

The **ROUND** function is a little more sophisticated. It uses the standard rules for rounding: if a number is less than 5 it rounds down, if it's 5 or more it rounds up. The syntax is as follows:

> ROUND(number, num_digits)
> number: the number to be rounded
> num_digits: the number of digits to round the number argument

For example, to round the number 3.14159 to two decimal places, enter in **=ROUND(3.14159,2)**.

There are two functions that allow you to control the direction of rounding. **ROUNDUP** will always round up, away from zero; **ROUNDDOWN** will always round down, toward zero. Both of these functions take the same arguments as **ROUND** above, so you can choose the number of digits to round to.

The **MROUND** function is useful in engineering because it rounds a number to a specified multiple. Its syntax is:

> MROUND(number, multiple)
> number: the number to be rounded
> multiple: the multiple to which the number will be rounded (i.e. 10, 5, 2, 0.25, etc.)

Worksheet 01k contains a flow calculation for a pipe with a known flowrate. The velocity in the pipe cannot exceed 120 in/sec. Calculations have already been done to determine the required diameter. However, pipe won't be available with a diameter of exactly 4.4 in, so we need round to the next available size.

In this case, it's necessary to round up to ensure the maximum velocity isn't exceeded. There are a few different ways to do this.

We can use the **INT** function, but this function will round down to 4. To avoid exceeding the desired velocity, simply add 1 to the pipe diameter. Enter **=INT(C9)+1** into the first cell for pipe diameter:

Inputs	
Flow Rate	475 GPM
Maximum Allowable Velocity	120 in/sec

Calculations	
Flow Rate	1829 in^3/sec
Area	15.2 in^2
Diameter	4.40 in

Result	
Pipe Diameter	=INT(C9+1)
Pipe Diameter	in
Pipe Diameter	in

The **ROUNDUP** function will ensure that we round up. To round to a whole number, enter 0 for the number of digits: **=ROUNDUP(C9,0)**

What if pipe is available in increments of half an inch? The **MROUND** function will allow us to round to the nearest half inch. Use 0.5 for the multiple: **=MROUND(C9,0.5)**

Inputs	
Flow Rate	475 GPM
Maximum Allowable Velocity	120 in/sec

Calculations	
Flow Rate	1829 in^3/sec
Area	15.2 in^2
Diameter	4.40 in

Result	
Pipe Diameter	5.00 in
Pipe Diameter	5.00 in
Pipe Diameter	=MROUND(C9,0.5)

This returns a value of 4.5, the nearest multiple of 0.5.

One thing to note is that the **ROUND** function can be used to round to the tens place, hundreds place, or any higher place. Simply use a negative number in the second argument – for the tens place, -1; for the hundreds place, -2; and so on. The magnitude of this number will be equal to the number of zeroes in your final rounded number.

For example, try rounding the flow rate to the hundreds place. Choose a cell and enter **=ROUND(C7,-2)**. To round to the tens place, use **-1** as the second argument. The **ROUNDUP** and

ROUNDDOWN functions work similarly: use a negative number as the second argument to round to a place left of the decimal.

Section 9: SUMIF and SUMIFS

Worksheets: 01m Time Above Threshold, 01n Time Between

This section and the next will cover the advanced summing functions available in Excel.

The **SUMIF** and **SUMIFS** functions enable you to sum values in a range of cells if they meet specified criteria. Values that don't meet the criteria are excluded from the sum.

The syntax for **SUMIF** is as follows:

SUMIF(range, criterion, [sumrange])
range: range of cells to evaluate
criterion: number, expression, function etc. that indicates which cells should be added
sumrange: (optional) the cells to add, if different from "range"

The criterion can be a number, an expression, a function or a text string. **SUMIF** restricts the data being summed according to a single criterion. **SUMIFS** allows you to specify multiple criteria. The syntax is:

SUMIFS(sum_range, criterion_range1, criterion1, [criterion_range2, criterion2], …)
sum_range: range of cells to add
criterion_range1: the range that is evaluated against criterion1
criterion1: number, expression, function etc. that indicates which cells in criterion_range 1 should be added
criterion_range2, criterion 2: (optional) additional criterion and the corresponding range

You may enter additional criteria as needed.

Worksheet 01m contains some measured stress data in column B as well as the number of times the stress was at a certain level during the measurement period in column C. We will use **SUMIF** to calculate the percent of time that the stress was above a certain level. In order to do that, you'll first calculate the percent of time represented by each count in column C. To do this, we'll divide each count by the sum of all the counts. Enter **=C6/SUM($C$6:$C$3)** in cell D6. If you prefer, you can select the cells with the mouse – remember to type **F4** after you select the range for all the counts to create an absolute reference. This way the values in the denominator won't change as we fill the formula into the rest of the column. **Select** cell D6 and **double-click** the fill handle.

Now column D displays the percentage of time that this location was at each stress level. In column G, we'll use **SUMIF** to calculate the percentage of time that the stress was greater than certain amounts. Enter **=SUMIF(** in cell G6. The range to be evaluated is in column B, so **select** that range (click the first cell and type **Ctrl-Shift-Down Arrow**), then type **F4** to make it an absolute reference.

So far you should have:

> **=SUMIF($B$6:$B$32**

Add a comma after the first argument. For the second argument, we have to build the criterion. We'll use the concatenate function (**&**) to form a string joining together the greater than symbol (**">"**) and the location of the cell that we're comparing against (**F6**). Therefore, the second argument will be: **">"&F6**. It's important to include the quotation marks.

At this point your formula will be:

> **=SUMIF($B$6:$B$32,">"&F6**

Again, add a comma after the argument. Lastly, we need to select the data to be summed – the percent data in column D. **Click** cell D6, type **Ctrl-Shift-Down Arrow**, followed by **F4**. Add a close parenthesis. Your formula should be:

> **=SUMIF($B$6:$B$32,">"&F6,$D$6:$D$32)**

Data		
Stress Level (psi)	Counts	Percent
1000	4834	6.8%
3000	4942	6.9%
5000	5336	7.5%
7000	4752	6.6%
9000	4410	6.2%
11000	4029	5.6%

Result	
Stress Level (psi)	Percent Time Greater
0	=SUMIF($B$6:$B$32,">"&F6,$D$6:$D$32)
10000	
20000	
30000	
40000	
50000	

This formula tells Excel to check if the value in column B is greater than the value in column F, and if so sum the corresponding percentages from column D.

Type **Enter**, **select** the cell again, and **double-click** the fill handle. This is the resulting table:

Result	
Stress Level (psi)	Percent Time Greater
0	100.0%
10000	66.1%
20000	40.3%
30000	20.1%
40000	6.9%
50000	0.6%

It's obvious from the data that the stress level is greater than zero 100% of the time. To check the other values, you can **highlight** the cells in column D that are greater than 10,000. Excel will automatically display the sum of the highlighted cells in the lower left border of the window:



The sum should match the value calculated by the **SUMIF** function.

Worksheet 01n contains the same data, but poses a different question: what percentage of the time was the stress between two values? We'll add all the percentage values between the two limits shown on the worksheet (6,000-20,000 psi), including the limits themselves. To answer this question, we'll solve use the **SUMIFS** function with two criteria, one for the lower limit and one for the upper limit. In the previous problem, we simply used the greater than (>) operator. In this example, we will include the limits in the criteria to see how to use the ≥ and ≤ operators in Excel.

You may have noticed that the **SUMIFS** function has a different order for its arguments. The range to be summed comes first. Enter **=SUMIFS(** in cell K5. **Select** the percentage data in column D.

The second argument is the range of the first criterion. In this case, the criterion will be based on the stress data in column B. At this point your formula should be:

=SUMIFS(D7:D33,B7:B33,

The third argument, the first criterion, will be constructed similarly to the first example, but instead of **“>”** we will use **“>=”** which is Excel’s form of the ≥ operator. Use **&** to concatenate to the cell containing the lower limit, G5.

=SUMIFS(D7:D33,B7:B33,”>=”&G5

That completes our first criterion. We also need to add a criterion to limit the summed values for stress levels less than 20,000 (cell G6). The SUMIFS function can take additional criteria by adding arguments for the range and criterion. For this problem, the range will be the same. The criterion will simply be **“<=”&G6**, restricting the summed values to only those with a stress less than or equal to G6.

The final formula will be:

=SUMIFS(D7:D33,B7:B33,“>=”&G5,B7:B33,“<=”&G6)



Again, you can quickly check that this formula summed only the values that fall within the limits by **highlighting** the percentage values that correspond to stress between 6,000 and 20,000 psi. The value in the lower border should match:

You can adjust the limits in cells G5 and G6 and the **SUMIFS** function will update accordingly.

Section 10: SUMPRODUCT and SUMSQ

Worksheet: 01o Angle Between Vectors

There are two more advanced summing functions that we'll discuss – **SUMPRODUCT** and **SUMSQ**.

SUMPRODUCT multiplies individual components in arrays together and then returns the sum of those components. The syntax is as follows:

SUMPRODUCT(array1, [array2], [array3],...)
array1: the first array whose components will be multiplied then added
array2, array 3, etc.: (optional) additional arrays whose components will be multiplied
then added

The first argument, array1, is the only one that is required. If only one argument is entered, the function will simply return the sum of the array components without doing any multiplication. To fully utilize this function, enter multiple arrays (up to 255). The function will multiply the individual array components and sum them.

SUMSQ returns the sum of squares of whatever numbers are entered as arguments. Its syntax is:

SUMSQ(number, [number2], [number3],...)
number1: first number for which the sum of squares is calculated
number2, number3, etc.: (optional) additional numbers for which the sum of squares will be
calculated

Again, only one argument is required. In that case, the function will return the value of that number squared. As you add additional values, they will be squared, and then summed together to return a single value.

Worksheet 01o contains an example that uses both functions. The spreadsheet contains XYZ coordinates for two vectors. We can calculate the angle between the vectors by using the equation:

$$\theta = \cos^{-1}\left(\frac{\vec{u}\cdot\vec{v}}{||\vec{u}||\cdot||\vec{v}||}\right)$$

This equation uses the dot product of the vectors to calculate the angle. The **SUMPRODUCT** function is essentially a dot product, so this function can be used to calculate the numerator. Enter **=SUMPRODUCT(** into cell C11. Select the XYZ coordinates for vector 1, then type a comma. Select the coordinates for vector 2. The resulting formula is:

=SUMPRODUCT(C6:E6,C7:E7)

Inputs			
Vector Coord's	X	Y	Z
Vector 1	4	3	0
Vector 2	2	8	0

Calculations	
Numerator	=SUMPRODUCT(C6:E6,C7:E7)
Denominator	

The denominator will be the product of the magnitude of the two vectors. To find the magnitude, take the square root of the sum of squares of the vector components. Enter in the formula:

=SQRT(SUMSQ(C6:E6))*SQRT(SUMSQ(C7:E7))

Finally, to calculate the angle in degrees, use the inverse cosine function, **ACOS**, inputting the numerator and denominator that was just calculated. To get the angle in degrees, nest the **ACOS** function inside of the **DEGREES** function:

=DEGREES(ACOS(C11/C12))

Inputs			
Vector Coord's	X	Y	Z
Vector 1	4	3	0
Vector 2	2	8	0

Result
=DEGREES(ACOS(C11/C12))

Calculations	
Numerator	32.00
Denominator	41.23

This formula will return an angle of 39.1 degrees.

Section 11: Statistical Functions

Worksheet: 01p Aluminum Ultimate Strength

Excel has many statistical functions. To see the complete list, go to **Formulas** > **More Functions** > **Statistical**. This section will cover some of the basic statistical functions.

Function	Description
AVERAGE	Finds the average or the mean of a set
MEDIAN	Finds the median of a set
MODE	Finds the mode of a set
STDEV	Finds the standard deviation of a set
MAX	Finds the maximum value in a set
MIN	Finds the minimum value in a set

Worksheet 01p contains measured ultimate strength of some samples of an aluminum alloy. We'll calculate some of the statistics for this data.

The first one we will calculate is the mean. In Excel, the function for the mean is **AVERAGE**. Enter that function, select the first cell of data, and type **Ctrl-Shift-Down Arrow**. This will give you the formula **=AVERAGE(B4:B103)**.

For the remaining cells, do likewise, using the functions in the table above.

Statistics	
Mean	45036
Median	45031
Mode	44958
St. Dev.	482.6
Maximum	46314
Minimum	44061

Section 12: CONCATENATE

Worksheet: 01q Text Functions

The **CONCATENATE** function is used to combine text data. It can connect multiple words or phrases into one longer phrase. There are two ways to use this function.

CONCATENATE can be used as a regular function. Input two or more arguments, and it will connect them together. The arguments can be a text value, a number, or a cell reference. Note that if you are using Excel 2016, you may use **CONCAT** instead; it works the same way.

In Worksheet 01q, there are two cells containing strings that we'd like to connect. Enter the formula **=CONCATENATE(C4,C5)**. This will connect the two text strings together:



However, **CONCATENATE** joins together the cells with no spaces or punctuation. We can add spaces and punctuations as arguments – just enclose them in double quotation marks so that Excel recognizes them as strings. To add a period and space after "Trust me" and a period after "I'm an Engineer," use this formula:

= CONCATENATE(C4,". ",C5,".")



However, we have already concatenated strings in previous sections using a different method: the ampersand (**&**). This operator allows you to simply connect strings rather than entering a function name and arguments. To get the second version above, complete with punctuation, enter in the formula:

= C4&". "&C5&"."

Section 13: LEFT, RIGHT, MID

Worksheet: 01q Text Functions

While the **CONCATENATE** function can *combine* text, Excel has other functions that can *separate* text into smaller strings. As an example, we'll separate the final text string from the previous section into two sentences.

The **LEFT** function extracts a certain number of characters of text starting from the left side. This function takes two arguments: the cell containing text, and the number of characters to extract.

Enter in the formula: **=LEFT(C4,9)**. This will return the first 9 characters of the string in cell C4.

Combined	Trust me. I'm an Engineer.
Text 1	Trust me.
Text 2	

There are two functions that we could use to obtain the second sentence. First, we'll use RIGHT. The syntax is the same as LEFT, except that it counts the number of characters starting from the right side of the string. In this case, the second sentence is 16 characters, so the formula will be: **=RIGHT(C4,16)**.

Combined	Trust me. I'm an Engineer.
Text 1	Trust me.
Text 2	I'm an Engineer.

The second sentence can also be extracted using the **MID** function. **MID** can return text from the middle of a string, but in this case, we'll use it to get the entire second sentence. The syntax of the **MID** function is slightly different, because you need to specify a starting position. In this example, the first character to be extracted is the 11^{th} character, counting from the left. We'll still extract 16 characters, so that will be the last argument. This function will be:

=MID(C4,11,16)

This will retrieve the second sentence.

Section 14: Single IF Function

Worksheet: 01r Safety factor

Excel has logical functions to help your spreadsheets make decisions. The most common logical function is the IF function. This function can be used to return a result that is dependent on whether an argument is true or false. The syntax is as follows:

IF(logical_test, [value_if_true], [value_if_false])

logical_test: the condition that will be tested

value_if_true: (optional) a value, function or string to return if the logical_test is true

value_if_false: (optional) a value, function or string to return if the logical_test is false

Worksheet 01r contains a simple calculation of tensile stress due to an axial force in a round shaft. The safety factor is then calculated based on the yield strength of the material. An **IF** function can be used to display a string based on the safety factor in cell C12. If the value is greater than 2, the **IF** function will return a text string that says the safety factor is acceptable. If it's less than 2, the function will return a string that says the safety value is not acceptable.

The first argument, logical_test, will be: **C12>2**. This is the argument that will be tested to see if it's true or not. If C12 is greater than 2, the function will return the value_if_true argument; if C12 is less than 2, it will return the value_if_false argument. The function will begin with: **=IF(C12>2**

The value_if_true will be the text string **"Safety Factor is Acceptable."** You can simply enter this as your second argument, after a comma, keeping the double quotation marks so Excel recognizes the input as a string.

The last argument, value_if_false, will simply be **"Safety Factor is NOT Acceptable."** This tells Excel what to return if the condition is not met. The complete function will be:

=IF(C12>2,"Safety Factor is Acceptable","Safety Factor is NOT Acceptable")

Tensile Stress

Inputs		
Force	1000	N
Diameter	5	mm
Yield Strength	372	MPa

Result
Safety Factor is Acceptable

Calculations		
Area	20	mm^2
Stress	50.9	MPa
Safety Factor	7.3	--

With the default values in the spreadsheet, the safety factor is 7.3, so this function will return the string “Safety Factor is Acceptable.” If you change the inputs, such as increasing the force to 100,000 N, the safety factor will drop below the threshold, and the IF function will return the string “Safety Factor is NOT Acceptable.”

Note the IF function can be used for outputs other than text strings. You can use it to return a number, do a calculation, or perform a function.